

*This program contains the information needed to create  Table 9 of the paper;
*The program references:

1) daily_panel_rfs which  is constructed in  Daily Panel RFS (Intermediate File) code
2) dd_posts18_21q2 which contains due dilligence posts from WSB. This data is collected from the pushshift API;


libname mylib 'g:\Dropbox\Wall Street Bets (Private)\Data';



DATA INFO;
SET MYLIB.daily_panel_rfs;

keep net_dd_posts ticker date cum_month gme_amc_flag post_gme
ret1 ret5 ret21 week1-week13
net_dd2  NET_DD2_POST NET_SA2 NET_SA2_POST NON_RESEARCH2 NON_RESEARCH2_POST
DD_and_SA DD_and_SA_POST DD_and_Non_Research DD_and_Non_Research_post net_dd2_pre
 ln_size ln_bm  abn_ret mom5 mom6_26     
BM_MISSING SIZE_MISSING   news_sentiment lag_sent5 lag_sent6_26 SENTIMENT21 total_dd_posts NON_RESEARCH_POSTS;
run;


proc sort data=info;
by ticker date;
run;

data info;
set info;
count +1;
by ticker;
if non_research_posts = . then non_research_posts =0;

l1_non_research = lag(non_research_posts);
l2_non_research = lag2(non_research_posts);
l3_non_research  = lag3(non_research_posts);
l4_non_research  = lag4(non_research_posts);
l5_non_research  = lag5(non_research_posts);
if count <=1 then l1_non_research = .;
if count <=2 then l2_non_research = .;
if count <=3 then l3_non_research = .;
if count <=4 then l4_non_research = .;
if count <=5 then l5_non_research = .;


past_posts = l1_non_research + l2_non_research + l3_non_research + l4_non_research + l5_non_research ;
if past_posts >1 then high_wsb_posts =1;
else high_wsb_posts =0;


lag_ret = lag(abn_ret);
abs_lag_ret = abs(lag_ret);

lag_abs_ret_rank =  abs_lag_ret;


run;



proc sort data=info;
by date;
run;
proc rank data=info out=info groups =100;
by date;
var lag_abs_ret_rank;
run;



data info;
set info;
if lag_abs_ret_rank >=90 then high_abs_ret =1;
else high_abs_ret = 0;
attention = max(high_abs_ret, high_wsb_posts);
run;


data pp_fund;
set mylib.dd_posts18_21q2;
date = day0;
if pp_num_words = . then delete;

if pp_num_words > funda_num_words then pp =1; else pp =0;

run;



proc sort data=pp_fund;
by ticker  date;
run;

proc univariate noprint data=pp_fund;
by ticker  date;
var purchase;
output out=nonpp_posts sum=buy_dd_posts_nonpp  N=total_dd_posts_nonpp;
where pp ~=1;
run;



data nonpp_posts;
set nonpp_posts;
sell_dd_posts_nonpp =  total_dd_posts_nonpp - buy_dd_posts_nonpp;
NET_DD_POSTS_nonpp = buy_dd_posts_nonpp - sell_dd_posts_nonpp;
 net_dd2_nonpp = NET_DD_POSTS_nonpp;
run;

PROC SORT DATA=nonpp_posts;
BY TICKER  date;
RUN;
PROC SORT DATA=INFO;
BY TICKER  date;
RUN;
DATA INFO;
MERGE INFO nonpp_posts;
BY TICKER  date;
if net_dd2_nonpp >6 then net_dd2_nonpp = 6;
if net_dd2_nonpp =  . then net_dd2_nonpp =0;
net_dd2_pp =   net_dd2 - net_dd2_nonpp;

net_dd2_pp_post = net_dd2_pp * post_gme;


if ret5 = . then delete;
run;



data info2;
set info;

pre_gme = (1- post_gme);


net_dd2_attention = net_dd2 * attention;
net_dd2_attention_post = net_dd2_attention * post_gme;






total_dd_posts2 = total_dd_posts;
if total_dd_posts2 >6 then total_dd_posts2 =6;


if net_dd2_pp >0 then pp_flag =1;
else pp_flag =0;





net_dd2_pp_flag = net_dd2 * pp_flag;
net_dd2_pp_flag_post = net_dd2_pp_flag * post_gme;


attention_pp = max(attention, pp_flag);




net_dd2_attention_pp = net_dd2 * attention_pp;
net_dd2_attention_pp_post = net_dd2 * post_gme * attention_pp;


attention_pp_post_gme = attention_pp * POST_GME;

run;



*code below generates Specification 1 of Table 9;
*  To obtain Specification 2, impose the filter "where gme_amc_flag  =0"

*To obtain Specification 3 replace "net_dd2_pp" and "net_dd2_pp_post" with "net_dd2_attention" and "net_dd2_attention_post";

*To obtain Specification 4 replace "net_dd2_pp" and "net_dd2_pp_post" with "net_dd2_attention" and "net_dd2_attention_post" and impose the filter "where gme_amc_flag  =0";

*To obtain Specification 5 replace "net_dd2_pp" and "net_dd2_pp_post" with "net_dd2_attention_pp" and "net_dd2_attention_pp_post";

*To obtain Specification 6 replace "net_dd2_pp" and "net_dd2_pp_post" with "net_dd2_attention_pp" and "net_dd2_attention_pp_post" and impose the filter "where gme_amc_flag  =0";



proc standard data=info2 out=info2_fe mean=0;
by date;
var ret21    net_dd2 net_dd2_post     net_dd2_attention_pp net_dd2_attention_pp_post
net_dd2_attention net_dd2_attention_post net_dd2_pp net_dd2_pp_post

NET_SA2 NET_SA2_POST NON_RESEARCH2 NON_RESEARCH2_POST
 ln_size ln_bm  abn_ret mom5 mom6_26       news_sentiment lag_sent5 lag_sent6_26   BM_MISSING SIZE_missing net_dd2_pp_flag net_dd2_pp_flag_post attention_pp_post_gme ;
 * where gme_amc_flag >=0;
quit;


%let yourdata=info2_fe;

%let firmid=ticker ;
%let time=cum_month;

%let y=ret21 ;
%let x=  net_dd2 net_dd2_post net_dd2_pp net_dd2_pp_post NON_RESEARCH2 NON_RESEARCH2_POST NET_SA2 NET_SA2_POST ln_size ln_bm  abn_ret mom5 mom6_26       news_sentiment lag_sent5 lag_sent6_26   BM_MISSING SIZE_MISSING;
RUN;

proc surveyreg data=&yourdata;

    cluster &firmid;
    model &y = &x /covb;
    ods output covb=firm;
run;
quit;


*cluster by second dimension (e.g., year);
proc surveyreg data=&yourdata;

    cluster &time;
    model &y = &x /covb;
    ods output covb=year;
run;
quit;

*cluster by intersection of the two dimensions (e.g, firm-year);
proc surveyreg data=&yourdata;
    cluster &firmid &time;
    model &y = &x /covb;
    ods output covb=both;
    ods output parameterestimates=parm;
run;
quit;

*keeps original parameter estimates;
data parm; set parm;
    keep parameter estimate;
run;

*returns a dataset with a scalar for the dimensions of the var/cov matrix. This is needed to extract the square roots of the diagonals later on;
data parm1; set parm;
    n=_n_;
    m=1;
    keep m n;
run;

data parm1; set parm1; by m;
    if last.m;
    keep n;
run;

*uses matrix algebra interface to construct Var-cov matrix and extract the standard errors;
proc iml;
    use both;
    read all var _num_ into Z;
    print Z;
    use firm;
    read all var _num_ into X;
    print X;
    use year;
    read all var _num_ into Y;
    print Y;
    use parm1;
    read all var _num_ into n;
    print n;

    B=X+Y-Z;
    C=I(n);
    D=J(n,1);
    E=C#B;
    F=E*D;
    G=F##.5;

    print B;
    print G;
    create b from G [colname='stderr']; ;
    append from G;
quit;

*creates a dataset called 'results' that contains the parameter estimates, the SE's, and the t-stats;
data results; merge parm B;
    tstat=estimate/stderr;
run;

proc print data=results;
run;




data spec3;
set results;
estimate3 = estimate  ;
obs = _n_;
keep parameter estimate3 obs;
run;
data spec3b;
set results;
estimate3 = tstat ;
obs = _n_ + .5;
keep parameter estimate3 obs;
run;
data spec3;
set spec3 spec3b;
run;
proc sort data=spec3;
by obs;
run;

